

## 1 Week Two: Iterating a single difference equation

You should work your way through this worksheet, answering questions and making notes where appropriate.

- You should finish this worksheet before the lab in week 3 — you will be given a new worksheet in week 3.
- You are *expected* to bring this worksheet to your lab in week 3.
  - You *will* be asked in week 3 to *demonstrate* that you have worked through the worksheet.
  - Students who can not demonstrate in week 3 that they have finished this worksheet will be *penalised* when their first maple assignment is marked.
- You may find it useful to bring this worksheet to labs from weeks 3 on.

Starting this week you will be writing Maple programmes. You will find it very *useful* to save these programs onto a disk which you bring to subsequent labs.

1. Use a text editor such as NotePad (Programs/Accessories/NotePad) to write your program.
2. Save your program onto a disk (or alternatively onto the C drive) as a *text* file.
3. To load your program into Maple enter `read "A:/file";` read file is the name of your program.

This week we will investigate the dynamics of the following problem.

**Example 1 (Ecology)** *The population of carp in a lake increases through natural growth by 25% per year. Its size in the year  $n = 0$  is 1100. If every year 300 carps are harvested, find an expression for the size of the population after year  $n$ .*

The difference equation for this problem is given by

$$x_{n+1} = 1.25x_n - 300, \quad x_0 = 1100. \quad (1)$$

In this problem the growth rate is 25%, i.e. the fractional growth rate is 0.25, and 300 carps are harvested every year. If we change the values of our parameters how would equation (1) change?

Let the fractional growth rate be  $g$  and suppose that the number of carp harvested is  $h$ . Equation (1) becomes

$$x_{n+1} = (1.0 + g)x_n - h, \quad x_0 = 1100. \quad (2)$$

The following Maple program solves equation (2). Note that Maple ignores all input after a # sign. It is a good programming style to add explanatory remarks to any program that you write.

```

# carp.maple (28.07.03)
# A simple maple program to iterate the first-order difference
# equation
#  $x_{n+1} = 1.25x_n - 300$ ;

finalyear := 10;    # The final value of 'n'.
growth := 0.25;    # the fractional growth rate per season.
harvest := 300;    # the number of carp culled each season.
year := n->n;      # define the 'time' variable.

f := x -> (1+growth)*x - harvest; #  $x_{n+1} = f(x_n)$ 

# Instead of using the variable  $x_n$  we will use the variable
# carp_n
carp := proc(n)      # define the values of  $x_n$  recursively.
    option remember; # Note that using the option remember causes the
    f(carp(n-1))     # previous values  $y(n-1)$  to be retained so that
    end:             # subsequent values may be based on them.

carp(0) := 1100;    # the initial number of carp in the lake.

# calculate the ordered pairs (carp_n,time_n) for n=0..finalyear

solution := [seq([year(n),carp(n)],n=0..finalyear)];
array(solution);   # display the points.

plot(solution,style=POINT,symbol=BOX,color=BLACK);

```

The output from this program is shown in figure 1. If your program generates an error, ask for help!




Figure 1: Output from carp.maple

### Question 1

1. What is the population size in year  $n = 2$ ?
2. After how many years is the population size reduced below 500?
3. What is the population size in year  $n = 12$ ?
  - (a) Is this answer mathematically correct?
  - (b) Is it physically correct?
  - (c) What does your answer mean physically?

**Question 2** Set the harvesting rate in your program to 200. What will happen to the size of the carp population as  $n \rightarrow \infty$ ? Is this realistic? Justify your answer.

Suppose that we want to compare the evolution of the population size over 10 generations for two values of the harvesting parameter: 260 and 280. How do we do this?

1. Set `harvest = 260` in `carp.maple`. At the end of your program add a new line  
`p1 := plot(solution,style=POINT,symbol=BOX,color=BLACK):`
2. Run your program. Now type in the following command  
`with(plots):`
3. Set `harvest = 280` in `carp.maple`. At the end of your program change  
`p1 := plot(solution,style=POINT,symbol=BOX,color=BLACK):`  
to  
`p2 := plot(solution,style=POINT,symbol=CROSS,color=BLACK):`
4. Run your program. Now type in the command  
`display(p1,p2);`

Your figure should look like this




Figure 2: Comparison of population size in the carp model:  $h = 260$  (box) and  $h = 280$  (cross)

Figure 2 neither has axis labels nor a title. If you want to include these you should *not* put them on the individual plots. You should add them to the `display` command.

**Question 3** Read `?plot[options]` and redraw figure 2, adding a title and axis labels. Commands that you should look at include `font`, `labels`, `labelfont`, `title`, `titlefont`.

**Note.** You add your title/axis commands within the `display(p1,p2)` command. Something like

```
display(p1,p2,labels=["x label","y label"])
```

**Question 4** Use the Maple help commands to answer these questions. Try `?plot` and `?plot[options]`

1. We have used `symbol=BOX` and `symbol=CROSS`. What other symbol options are valid?
2. We have set `color=BLACK`. How many other color options are there?

**Question 5** A harvesting strategy is **unsustainable** if the size of the harvested species ever becomes zero, or negative. A harvesting strategy is **sustainable** if the population always remain strictly positive.

1. What is the highest value of  $h$  at which harvesting is sustainable in the carp model?
2. A fisheries manager asks you to recommend a value of  $h$  (the larger the value of  $h$ , the more profit). What value do you recommend? Justify your answer.

**Question 6**

1. Using the maple help command explain what the command `ceil` does.
2. Change the definition of the function  $f$  from  
 $f := x \rightarrow (1 + \text{growth}) * x - \text{harvest};$  to  
 $f := x \rightarrow \text{ceil}((1 + \text{growth}) * x - \text{harvest});$ 
  - (a) Repeat question 1 & 5.
  - (b) Explain the significance of the change you have made to your code and how it changes the output.

**Question 7** In 2003 an astute student pointed out during the lecture that we have assumed harvesting occurs **after** growth. If instead harvesting occurs **before** growth equation (2) becomes

$$x_{n+1} = (1 + g)(x_n - h) \tag{3}$$

What is the maximum sustainable harvest for equation (3) (take  $g = 0.25$ )?